
 Change 1
	Test case name
	8.5.4.1

	Reason for change
	A UE not supporting feature group indicator 8 (FGI8) may send either of the choice elements under “v390NonCriticalExtensions” so a different template need to be maintained to handle no support of FGI8.

	Summary of change
	Added handling of FGI8 and created and used new template cr_UeCapabilitiesRAT_Container_UTRAN_noFeatGrp_8

	Source of change
	UTRAN_CapabilityFunctions.ttcn

	MCC160 Comment
	Partly rejected
The access stratum release indicator is mandatory. See 25.331 cl.10.3.3.42. It is carried in the REL-4 extensions of the Radio Access Capability IE. See ASN.1 of 25.331. Therefore the choice elements under “v390NonCriticalExtensions” cannot be absent.
Partly accepted. 

However, there are IEs under “v390NonCriticalExtensions” which are mandatory only if some condition is met. All such IEs may be absent unless a related condition is satisfied.
Therefore these IEs my be either present or omitted. 
This checks the nature of the IEs as per ASN.1 specification. 

IEs becoming mandatory depending upon some condition being fulfilled are checked outside the matching functionality. See f_U_DecodeAndCheckInterRatHandoverInfo.


Before change:

	    function f_U_DecodeAndCheckInterRatHandoverInfo ( octetstring p_Octetstring ) return boolean

  { //36.331 cl. 6.3.6: The octet string contains the INTER RAT HANDOVER INFO message defined in TS 25.331

    var boolean v_MatchResult;

    var InterRATHandoverInfo v_Received_UeCapabilitiesRAT_Container_UTRAN := f_U_DecodeInterRatHandoverInfo (p_Octetstring); // @sic R5-120764 sic@

    var template UE_SecurityInformation2 v_UE_SecurityInformation2 := omit;

    if (pc_FeatrGrp_8) { // @sic R5s110350 sic@

      v_UE_SecurityInformation2 := cr_UE_SecurityInformation2;

    }

    v_MatchResult := match ( v_Received_UeCapabilitiesRAT_Container_UTRAN, cr_UeCapabilitiesRAT_Container_UTRAN (v_UE_SecurityInformation2));

    return v_MatchResult;

  }


After change:

	  function f_U_DecodeAndCheckInterRatHandoverInfo ( octetstring p_Octetstring ) return boolean

  { //36.331 cl. 6.3.6: The octet string contains the INTER RAT HANDOVER INFO message defined in TS 25.331

    var boolean v_MatchResult;

    var InterRATHandoverInfo v_Received_UeCapabilitiesRAT_Container_UTRAN := f_U_DecodeInterRatHandoverInfo (p_Octetstring); // @sic R5-120764 sic@

    var template UE_SecurityInformation2 v_UE_SecurityInformation2 := omit;

    if (pc_FeatrGrp_8) { // @sic R5s110350 sic@

      v_UE_SecurityInformation2 := cr_UE_SecurityInformation2;

      v_MatchResult := match ( v_Received_UeCapabilitiesRAT_Container_UTRAN, cr_UeCapabilitiesRAT_Container_UTRAN (v_UE_SecurityInformation2));

    }

    else

    {

    v_MatchResult := match ( v_Received_UeCapabilitiesRAT_Container_UTRAN, cr_UeCapabilitiesRAT_Container_UTRAN_noFeatGrp_8 (v_UE_SecurityInformation2));

    }   

    return v_MatchResult;

  }  


New template:

	  template InterRATHandoverInfo cr_UeCapabilitiesRAT_Container_UTRAN_noFeatGrp_8 (template UE_SecurityInformation2 p_UE_SecurityInformation2 := omit) :=

   {

   /* Values according to 36.523-1 Table 8.5.4.1.3.3-5: ueCapabilitiesRAT-Container-UTRAN */

    /* @status    APPROVED */

    predefinedConfigStatusList := {

      absent := NULL

    },

    uE_SecurityInformation := ?,

    ue_CapabilityContainer := {

      present_ := ?

    },

    v390NonCriticalExtensions := (

    {absent := NULL},

      {present_ := {

        interRATHandoverInfo_v390ext := ?,

         v3a0NonCriticalExtensions := ({

          interRATHandoverInfo_v3a0ext := ?,

          laterNonCriticalExtensions := {

            interRATHandoverInfo_v3d0ext := ?,

            interRATHandoverInfo_r3_add_ext := *,  

            v3g0NonCriticalExtensions := {

              interRATHandoverInfo_v3g0ext := ?,

              v4b0NonCriticalExtensions := {

                interRATHandoverInfo_v4b0ext := ?,

                v4d0NonCriticalExtensions := {

                  interRATHandoverInfo_v4d0ext := ?,

                  v590NonCriticalExtensions := {

                    interRATHandoverInfo_v590ext := ?,

                    v690NonCriticalExtensions := {

                      interRATHandoverInfo_v690ext := {

                        ue_SecurityInformation2 := p_UE_SecurityInformation2, // @sic R5s110350 sic@

                        ue_RadioAccessCapabilityComp := *,

                        ue_RadioAccessCapabilityComp2 := ?

                      },

                      v6b0NonCriticalExtensions := * // @sic R5s110244 sic@

                    }

                  }

                }

              }

            }

          }

        },omit)

      }

      } 

    )

  };  


Change 2
	Test case name
	8.5.4.1

	Reason for change
	A UE may or may not send IE interRATHandoverInfo_r3_add_ext as in this there is no IE which conatins information related to EUTRA. This IE contains other RAT (UTRA) information only.

	Summary of change
	Replaced “?” with “*” for IE interRATHandoverInfo_r3_add_ext inside "cr_UeCapabilitiesRAT_Container_UTRAN" template.

	Source of change
	UTRAN_CapabilityFunctions.ttcn

	MCC160 Comment
	Accepted.


Before change:

	  template InterRATHandoverInfo cr_UeCapabilitiesRAT_Container_UTRAN (template UE_SecurityInformation2 p_UE_SecurityInformation2 := omit) :=

  { /* Values according to 36.523-1 Table 8.5.4.1.3.3-5: ueCapabilitiesRAT-Container-UTRAN */

    /* @status    APPROVED */

    predefinedConfigStatusList := {

      absent := NULL

    },

    uE_SecurityInformation := ?,

    ue_CapabilityContainer := {

      present_ := ?

    },

    v390NonCriticalExtensions := {

      present_ := {

        interRATHandoverInfo_v390ext := ?,

        v3a0NonCriticalExtensions := {

          interRATHandoverInfo_v3a0ext := ?,

          laterNonCriticalExtensions := {

            interRATHandoverInfo_v3d0ext := ?,

            interRATHandoverInfo_r3_add_ext := ?,

…
…
  };


After change:

	  template InterRATHandoverInfo cr_UeCapabilitiesRAT_Container_UTRAN (template UE_SecurityInformation2 p_UE_SecurityInformation2 := omit) :=

  { /* Values according to 36.523-1 Table 8.5.4.1.3.3-5: ueCapabilitiesRAT-Container-UTRAN */

    /* @status    APPROVED */

    predefinedConfigStatusList := {

      absent := NULL

    },

    uE_SecurityInformation := ?,

    ue_CapabilityContainer := {

      present_ := ?

    },

    v390NonCriticalExtensions := {

      present_ := {

        interRATHandoverInfo_v390ext := ?,

        v3a0NonCriticalExtensions := {

          interRATHandoverInfo_v3a0ext := ?,

          laterNonCriticalExtensions := {

            interRATHandoverInfo_v3d0ext := ?,

            interRATHandoverInfo_r3_add_ext := *,
…
…
  };


MCC160 Implementation
The changes described above are implemented.

Additionally the access stratum release indicator is exploited and the UTRAN release version is checked (FDD: >Rel-7, TDD: > Rel-8).

Settings depending upon special conditions are done outside the matching. 

In addition the template cr_UeCapabilitiesRAT_Container_UTRAN is expanded as per the test case prose.

Before:

  template InterRATHandoverInfo cr_UeCapabilitiesRAT_Container_UTRAN (template UE_SecurityInformation2 p_UE_SecurityInformation2 := omit) :=
  { /* Values according to 36.523-1 Table 8.5.4.1.3.3-5: ueCapabilitiesRAT-Container-UTRAN */

    /* @status    APPROVED */

    predefinedConfigStatusList := {

      absent := NULL

    },

    uE_SecurityInformation := ?,

    ue_CapabilityContainer := {

      present_ := ?

    },

    v390NonCriticalExtensions := {

      present_ := {

        interRATHandoverInfo_v390ext := ?,

        v3a0NonCriticalExtensions := {

          interRATHandoverInfo_v3a0ext := ?,

          laterNonCriticalExtensions := {

            interRATHandoverInfo_v3d0ext := ?,

            interRATHandoverInfo_r3_add_ext := ?,

            v3g0NonCriticalExtensions := {

              interRATHandoverInfo_v3g0ext := ?,

              v4b0NonCriticalExtensions := {

                interRATHandoverInfo_v4b0ext := ?,

                v4d0NonCriticalExtensions := {

                  interRATHandoverInfo_v4d0ext := ?,

                  v590NonCriticalExtensions := {

                    interRATHandoverInfo_v590ext := ?,

                    v690NonCriticalExtensions := {

                      interRATHandoverInfo_v690ext := {

                        ue_SecurityInformation2 := p_UE_SecurityInformation2, // @sic R5s110350 sic@

                        ue_RadioAccessCapabilityComp := *,

                        ue_RadioAccessCapabilityComp2 := ?

                      },

                      v6b0NonCriticalExtensions := * // @sic R5s110244 sic@

                    }

                  }

                }

              }

            }

          }

        }

      }

    }

  };

After:

  template InterRATHandoverInfo cr_UeCapabilitiesRAT_Container_UTRAN :=

  { /* Values according to 36.523-1 Table 8.5.4.1.3.3-5: ueCapabilitiesRAT-Container-UTRAN */

    /* @status    APPROVED */

    predefinedConfigStatusList := {

      absent := NULL

    },

    uE_SecurityInformation := ?,

    ue_CapabilityContainer := {

      present_ := ?

    },

    v390NonCriticalExtensions := {

      present_ := {

        interRATHandoverInfo_v390ext := ?,

        v3a0NonCriticalExtensions := {

          interRATHandoverInfo_v3a0ext := ?,

          laterNonCriticalExtensions := {

            interRATHandoverInfo_v3d0ext := ?,

            interRATHandoverInfo_r3_add_ext := *, // @sic R5s120706 sic@
            v3g0NonCriticalExtensions := {

              interRATHandoverInfo_v3g0ext := ?,

              v4b0NonCriticalExtensions := {

                interRATHandoverInfo_v4b0ext := ?, // Access Stratum Release Indicator

                v4d0NonCriticalExtensions := ({

                  interRATHandoverInfo_v4d0ext := ?,

                  v590NonCriticalExtensions := ({

                    interRATHandoverInfo_v590ext := ?,

                    v690NonCriticalExtensions := ({

                      interRATHandoverInfo_v690ext := ?,

                      v6b0NonCriticalExtensions := ({ // @sic R5s110244 sic@

                        interRATHandoverInfo_v6b0ext := ?,

                        v6e0NonCriticalExtensions := ({

                          interRATHandoverInfo_v6e0ext := ?,

                          v770NonCriticalExtensions := ({

                            interRATHandoverInfo_v770ext := ?,

                            v790nonCriticalExtensions := ({

                              interRATHandoverInfo_v790ext := ?,

                              v860NonCriticalExtensions := ({

                                interRATHandoverInfo_v860ext := ?,

                                v880NonCriticalExtensions := ({

                                  interRATHandoverInfo_v880ext := ?,

                                  v920NonCriticalExtensions := ({

                                    interRATHandoverInfo_v920ext := ?,

                                    v8b0NonCriticalExtensions := ({

                                      interRATHandoverInfo_v8b0ext := ?,

                                      v950NonCriticalExtensions := ({

                                        interRATHandoverInfo_v950ext := ?,

                                        va40NonCriticalExtensions := ({

                                          interRATHandoverInfo_va40ext := ?,

                                          nonCriticalExtensions := *

                                        }, omit)

                                      }, omit)

                                    }, omit)

                                  }, omit)

                                }, omit)

                              }, omit)

                            }, omit)

                          }, omit)

                        }, omit)

                      }, omit)

                    }, omit)

                  }, omit)

                }, omit)
              }

            }

          }

        }

      }

    }

  };
Before
  function f_U_DecodeAndCheckInterRatHandoverInfo ( octetstring p_Octetstring ) return boolean

  { //36.331 cl. 6.3.6: The octet string contains the INTER RAT HANDOVER INFO message defined in TS 25.331

    var boolean v_MatchResult;

    var InterRATHandoverInfo v_Received_UeCapabilitiesRAT_Container_UTRAN := f_U_DecodeInterRatHandoverInfo (p_Octetstring); // @sic R5-120764 sic@

    var template UE_SecurityInformation2 v_UE_SecurityInformation2 := omit;

    if (pc_FeatrGrp_8) { // @sic R5s110350 sic@

      v_UE_SecurityInformation2 := cr_UE_SecurityInformation2;

    }

    v_MatchResult := match ( v_Received_UeCapabilitiesRAT_Container_UTRAN, cr_UeCapabilitiesRAT_Container_UTRAN (v_UE_SecurityInformation2));

    return v_MatchResult;

  }

After
  function f_U_DecodeAndCheckInterRatHandoverInfo ( octetstring p_Octetstring ) return boolean

  {

    var boolean v_MatchResult;

    var InterRATHandoverInfo v_Received_UeCapabilitiesRAT_Container_UTRAN := f_U_DecodeInterRatHandoverInfo (p_Octetstring); // @sic R5-120764 sic@

    var AccessStratumReleaseIndicator v_AccessStratumReleaseIndicator;
    if (isvalue(v_Received_UeCapabilitiesRAT_Container_UTRAN.v390NonCriticalExtensions.present_.v3a0NonCriticalExtensions.laterNonCriticalExtensions.v3g0NonCriticalExtensions.v4b0NonCriticalExtensions.interRATHandoverInfo_v4b0ext.accessStratumReleaseIndicator)) {

       v_AccessStratumReleaseIndicator:= v_Received_UeCapabilitiesRAT_Container_UTRAN.v390NonCriticalExtensions.present_.v3a0NonCriticalExtensions.laterNonCriticalExtensions.v3g0NonCriticalExtensions.v4b0NonCriticalExtensions.interRATHandoverInfo_v4b0ext.accessStratumReleaseIndicator;

    }

    else {

      f_UTRAN_SetVerdictInconc(__FILE__, __LINE__, "Access Stratum Release Indicator not provided although mandatory");

      // 25.331 cl. 8.1.16.3 regarding Radio Access capability

      // 25.331 cl. 10.3.3.42

      // -- UE-RadioAccessCapability is compatible with R99, although accessStratumReleaseIndicator

      // -- is removed from this IE, since its encoding did not result in bits. The

      // -- accessStratumReleaseIndicator is provided in the relevant REL-4 extension IEs.

    }
    // checking the UTRAN release version
    if (pc_FDD) { // release must Rel8 or greater

      if ((v_AccessStratumReleaseIndicator == rel_4) or

          (v_AccessStratumReleaseIndicator == rel_5) or

          (v_AccessStratumReleaseIndicator == rel_6) or

          (v_AccessStratumReleaseIndicator == rel_7)) {

        f_UTRAN_SetVerdictInconc(__FILE__, __LINE__, "UE Rel. version incorrect for FDD");

      }

    }
    if (pc_TDD_LCR or pc_TDD_HCR or pc_TDD_VHCR){ // release must Rel9 or greater

      if ((v_AccessStratumReleaseIndicator == rel_4) or

          (v_AccessStratumReleaseIndicator == rel_5) or

          (v_AccessStratumReleaseIndicator == rel_6) or

          (v_AccessStratumReleaseIndicator == rel_7) or

          (v_AccessStratumReleaseIndicator == rel_8)) {

        f_UTRAN_SetVerdictInconc(__FILE__, __LINE__, "UE Rel. version incorrect for TDD");

      }

    }
    if (pc_FeatrGrp_8) { // @sic R5s110350 sic@

      if (not (isvalue(v_Received_UeCapabilitiesRAT_Container_UTRAN.v390NonCriticalExtensions.present_.v3a0NonCriticalExtensions.laterNonCriticalExtensions.v3g0NonCriticalExtensions.v4b0NonCriticalExtensions.v4d0NonCriticalExtensions.v590NonCriticalExtensions.v690NonCriticalExtensions.interRATHandoverInfo_v690ext.ue_SecurityInformation2))) {

         f_UTRAN_SetVerdictInconc(__FILE__, __LINE__, "ue_SecurityInformation2 not present for Feature Group 8");

      }

    }
    if (pc_FDD) { // 25.331 cl. 8.1.16.3 regarding Radio Access capability

      if (not (isvalue(v_Received_UeCapabilitiesRAT_Container_UTRAN.v390NonCriticalExtensions.present_.v3a0NonCriticalExtensions.laterNonCriticalExtensions.v3g0NonCriticalExtensions.v4b0NonCriticalExtensions.v4d0NonCriticalExtensions.v590NonCriticalExtensions.v690NonCriticalExtensions.interRATHandoverInfo_v690ext.ue_RadioAccessCapabilityComp2))) {

         f_UTRAN_SetVerdictInconc(__FILE__, __LINE__, "ue_RadioAccessCapabilityComp2 not present for FDD");

      }

    }
    if (pc_TDD_LCR) { // 25.331 cl. 8.1.16.3 regarding Radio Access capability

      if (not (isvalue(v_Received_UeCapabilitiesRAT_Container_UTRAN.v390NonCriticalExtensions.present_.v3a0NonCriticalExtensions.laterNonCriticalExtensions.v3g0NonCriticalExtensions.v4b0NonCriticalExtensions.v4d0NonCriticalExtensions.v590NonCriticalExtensions.v690NonCriticalExtensions.v6b0NonCriticalExtensions.v6e0NonCriticalExtensions.v770NonCriticalExtensions.v790nonCriticalExtensions.v860NonCriticalExtensions.v880NonCriticalExtensions.interRATHandoverInfo_v880ext.ue_RadioAccessCapabilityInfo_TDD128))) {

         f_UTRAN_SetVerdictInconc(__FILE__, __LINE__, "ue_RadioAccessCapabilityInfo_TDD128 not present for TDD_LCR");

      }

      if (not (isvalue(v_Received_UeCapabilitiesRAT_Container_UTRAN.v390NonCriticalExtensions.present_.v3a0NonCriticalExtensions.laterNonCriticalExtensions.v3g0NonCriticalExtensions.v4b0NonCriticalExtensions.v4d0NonCriticalExtensions.v590NonCriticalExtensions.v690NonCriticalExtensions.v6b0NonCriticalExtensions.v6e0NonCriticalExtensions.v770NonCriticalExtensions.v790nonCriticalExtensions.v860NonCriticalExtensions.v880NonCriticalExtensions.v920NonCriticalExtensions.v8b0NonCriticalExtensions.interRATHandoverInfo_v8b0ext.ue_RadioAccessCapabilityInfo))) {

         f_UTRAN_SetVerdictInconc(__FILE__, __LINE__, "ue_RadioAccessCapabilityInfo not present for TDD_LCR");

      }

    }
    v_MatchResult := match ( v_Received_UeCapabilitiesRAT_Container_UTRAN, cr_UeCapabilitiesRAT_Container_UTRAN );

  } // f_U_DecodeAndCheckInterRatHandoverInfo

